Includes barcode for easy software entry.

HEWLETT-PACKARD

HP-41

USERS' LIBRARY SOLUTIONS

Electrical Engineering



NOTICE

The program material contained herein is supplied without representation or warranty of any kind. Hewlett-Packard Company therefore assumes no responsibility and shall have no liability, consequential or otherwise, of any kind arising from the use of this program material or any part thereof.

INTRODUCTION

This HP-41C Solutions book was written to help you get the most from your calculator. The programs were chosen to provide useful calculations for many of the common problems encountered.

They will provide you with immediate capabilities in your everyday calculations and you will find them useful as guides to programming techniques for writing your own customized software. The comments on each program listing describe the approach used to reach the solution and help you follow the programmer's logic as you become and expert on your HP calculator.

KEYING A PROGRAM INTO THE HP-41C

There are several things that you should keep in mind while you are keying in programs from the program listings provided in this book. The output from the HP 82143A printer provides a convenient way of listing and an easily understood method of keying in programs without showing every keystroke. This type of output is what appears in this handbook. Once you understand the procedure for keying programs in from the printed listings, you will find this method simple and fast. Here is the procedure:

1. At the end of each program listing is a listing of status information required to properly execute that program. Included is the SIZE allocation required. Before you begin keying in the program, press [XEQ] [ALPHA] SIZE [ALPHA] and specify the allocation (three digits; e.g., 10 should be specified as 010).

   Also included in the status information is the display format and status of flags important to the program. To ensure proper execution, check to see that the display status of the HP-41C is set as specified and check to see that all applicable flags are set or clear as specified.

2. Set the HP-41C to PRGM mode (press the [PRGM] key) and press ■ [GTO] [•] [•] to prepare the calculator for the new program.

3. Begin keying in the program. Following is a list of hints that will help you when you key in your programs from the program listings in this handbook.

   a. When you see " (quote marks) around a character or group of characters in the program listing, those characters are ALPHA. To key them in, simply press [ALPHA], key in the characters, then press [ALPHA] again. So "SAMPLE" would be keyed in as [ALPHA] "SAMPLE" [ALPHA].

   b. The diamond in front of each LBL instruction is only a visual aid to help you locate labels in the program listings. When you key in a program, ignore the diamond.

   c. The printer indication of divide sign is /. When you see / in the program listing, press [÷].

   d. The printer indication of the multiply sign is *. When you see * in the program listing, press [×].

   e. The ⊢ character in the program listing is an indication of the [APPEND] function. When you see ⊢, press ■ [APPEND] in ALPHA mode (press ■ and the K key).

   f. All operations requiring register addresses accept those addresses in these forms:

      nn (a two-digit number)
      IND nn (INDIRECT: ■ , followed fy a two-digit number)
      X, Y, Z, T, or L (a STACK address: [•] followed by X, Y, Z, T, or L)
      IND X, Y, Z, T or L (INDIRECT stack: ■ [•] followed by X, Y, Z, T, or L)

      Indirect addresses are specified by pressing ■ and then the indirect address. Stack addresses are specified by pressing [•] followed by X, Y, Z, T, or L. Indirect stack addresses are specified by pressing ■ [•] and X, Y, Z, T, or L.

Printer Listing	Keystrokes	Display
01♦LBL "SAMPLE"	■ [LBL] [ALPHA] SAMPLE [ALPHA]	*01 LBL^{T}SAMPLE*
02 "THIS IS A "	[ALPHA] *THIS IS A* [ALPHA]	*02^{T}THIS IS A*
03 "⊢SAMPLE"	[ALPHA] ■ [APPEND] SAMPLE	*03^T ⊢ SAMPLE*
04 AVIEW	■ AVIEW [ALPHA]	*04 AVIEW*
05 6	6	*05 6*
06 ENTER↑	[ENTER↑]	*06 ENTER ↑*
07 -2	2 [CHS]	*07 -2*
08 /	[÷]	*08 /*
09 ABS	[XEQ] [ALPHA] ABS [ALPHA]	*09 ABS*
10 STO IND L	[STO] ■ [•] L	*10 STO IND L*
11 "R3="	[ALPHA] R3= ■ [ARCL] 03	*11^{T}R3=*
12 ARCL 03	■ [AVIEW]	*12 ARCL 03*
13 AVIEW	[ALPHA]	*13 AVIEW*
14 RTN	■ [RTN]	*14 RTN*

TABLE OF CONTENTS

RC TIMING

This program computes any one of the six variables shown, provided the other five are known.





V_1 = Voltage before step	C = Capacitance
V_2 = Voltage after step	Vi = Instantaneous voltage
R = Resistance	t = Time

All solutions are algebraically derived from this basic formula:

$$Vi = V_1 e^{-t/RC} + V_2(1 - e^{-t/RC})$$

NOTE: For voltages across the resistor, remember that $V_R + V_C = V$ applies at all times.

Example:

A 555 type of integrated circuit timer uses an external RC configuration for time determination. When used as a one-shot its output pulse terminates when the capacitor charges to 2/3 of the supply voltage. Until the pulse starts, the capacitor is shorted so $V_1 = 0$. Given a supply voltage of 12V, a 47μF capacitor, and you need a 1 second pulse, what size resistor should you use?

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 007	
[XEQ] [ALPHA] RC [ALPHA]	V1?
0 [R/S]	V2?
12 [R/S]	VI?
8 [R/S]	C?
47 [EEX] 6 [CHS] [R/S]	R?
[R/S]	T?
1 [R/S]	1.000 00
[XEQ] [ALPHA] R [ALPHA]	R=19.37E3

User Instructions

SIZE: 007

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	Begin execution and enter data, skip		[XEQ] RC	V1?
	unknown by pressing [R/S].	V_1	[R/S]	V2?
		V_2	[R/S]	VI?
		V_i	[R/S]	C?
		C	[R/S]	R?
		R	[R/S]	T?
		t	[R/S]	
3	Execute for desired output:			
	V_1		[XEQ] V1	V1=
	V_2		[XEQ] V2	V2=
	V_i		[XEQ] VI	VI=
	C		[XEQ] C	C=
	R		[XEQ] R	R=
	t		[XEQ] T	T=
4	To vary one or more parameters simply			
	place the new value in the appropriate			
	register	V_1	[STO] 01	
		V_2	[STO] 02	
		V_i	[STO] 03	
		C	[STO] 04	
		R	[STO] 05	
		t	[STO] 06	
	Go to step 3			

Program Listings

Initialize registers

```
◆LBL "RC"
ENG 3
CF 00
"V1?"
PROMPT
STO 01
"V2?"
PROMPT
STO 02
"VI?"
PROMPT
STO 03
"C?"
PROMPT
STO 04
"R?"
PROMPT
STO 05
"T?"
PROMPT
STO 06
STOP
```

V_2

```
◆LBL "V2"
XEQ 03
RCL 01
*
RCL 03
X<>Y
-
STO 00
XEQ 00
RCL 00
X<>Y
/
STO 02
"V2="
GTO 05
```

V_i

```
◆LBL "VI"
XEQ 03
RCL 01
*
STO 00
XEQ 00
RCL 02
*
RCL 00
+
STO 03
"VI="
GTO 05
```

V_1

```
◆LBL "V1"
XEQ 00
RCL 02
*
RCL 03
X<>Y
-
STO 00
XEQ 03
RCL 00
X<>Y
/
STO 01
"V1="
GTO 05
```

C

```
◆LBL C
XEQ 01
RCL 05
*
RCL 06
X<>Y
/
STO 04
"C="
GTO 05
```

R

```
◆LBL "R"
XEQ 01
RCL 04
*
RCL 06
X<>Y
/
STO 05
"R="
GTO 05
```

t

```
◆LBL "T"
XEQ 01
RCL 04
RCL 05
*
*
STO 06
"T="
◆LBL 05
ARCL X
AVIEW
STOP
```

```
◆LBL 01
1
RCL 03
RCL 01
-
```

Program Listings

```
103 RCL 02
104 RCL 01
105 -
106 /
107 -
108 LN
109 CHS
110 RTN
111+LBL 00
112 SF 05
113 1
114+LBL 03
115 RCL 06
116 RCL 05
117 RCL 04
118 *
119 /
120 CHS
121 E↑X
122 FS?C 05
123 -
124 RTN
125 .END.
```

$-\ln\left[1 - \frac{V_i - V_1}{V_2 - V_1}\right]$

$e^{-t/RC}$ or $1 - e^{-t/RC}$

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	Temporary Storage	50	
	V_1		
	V_2		
	V_i		
	C		
05	R	55	
	t		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 007 TOT. REG. 37 USER MODE

ENG 3 FIX ____ SCI ____ ON ____ OFF X

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
05		To calculate $1 - e^{-t/RC}$	To calculate $e^{-t/RC}$

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

RC TIMING

HEWLETT PACKARD
SOLUTION BOOK:
ELECTRICAL ENGINEERING

PROGRAM REGISTERS NEEDED: 31

ROW 1 (1 - 4)

ROW 2 (4 - 10)

ROW 3 (11 - 19)

ROW 4 (19 - 24)

ROW 5 (24 - 34)

ROW 6 (35 - 38)

ROW 7 (39 - 47)

ROW 8 (48 - 51)

ROW 9 (52 - 60)

ROW 10 (61 - 67)

ROW 11 (67 - 76)

ROW 12 (76 - 83)

ROW 13 (84 - 87)

ROW 14 (88 - 97)

ROW 15 (98 - 110)

ROW 16 (111 - 122)

ROW 17 (122 - 125)

FREQUENCY RESPONSE OF A TRANSFER FUNCTION

For transfer function of the form:

$$G\ (S) = \frac{K_1(Z_2S + 1)}{S^{N_3}(Z_4S + 1)\ (Z_5S + 1)\ (\frac{S^2}{\omega_7^2} + \frac{2Z_6S}{\omega_7} + 1)}$$

the program computes $\angle G\ (j\omega)$, $|G(j\omega)|$ and log $|G(j\omega)|$ for any input frequency ω.

Parameters K_1, Z_2, N_3, Z_4, Z_5, Z_6 and ω_7 are stored in registers 01, 02, 03, 04, 05, 06, and 07 respectively.

NOTE: For type 0 systems, enter $N_3 = 0$. Z_2, Z_4 and/or Z_5 can be entered as 0. If there is no quadratic term, enter Z_6 as 0 and ω_7 very large compared to $\frac{1}{Z_5}$, where Z_5 is the smallest first order term used (other than zero).

Example:

Find $|G(j\omega)|$, $\angle G(j\omega)$ and log $|G(j\omega)|$ for $G(S) = \frac{12(S + 0.6)}{S(S + 1)(S^2 + 6S + 36)}$ frequency 0.01 rad/sec.

First put G(S) into proper form: $G(S) = \frac{.2(1.67S + 1)}{S(S + 1)[(\frac{S}{6})^2 + (\frac{S}{6}) + 1]}$

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 010	
[XEQ] [ALPHA] FREQ [ALPHA]	K1=?
.2 [R/S]	Z2=?
1.67 [R/S]	N3=?
1 [R/S]	Z4=?
1 [R/S]	Z5=?
0 [R/S]	Z6=?
.5 [R/S]	W7=?
6 [R/S]	W=?
.01 [R/S]	G∠JW⅃=20.00
[R/S]	∠G∠JW⅃=-89.71
[R/S]	LOG G=1.30

User Instructions

SIZE: 010

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load program.			
2.	Begin Execution, enter data.		[XEQ] FREQ	K1=?
		K_1	[R/S]	Z2=?
		Z_2	[R/S]	N3=?
		N_3	[R/S]	Z4=?
		Z_4	[R/S]	Z5=?
		Z_5	[R/S]	Z6=?
		Z_6	[R/S]	W7=?
		ω_7	[R/S]	W=?
3.	Answer is automatic.	ω	[R/S]	G <JW> =
			[R/S]	∠G <JW> =
			[R/S]	LOG G <JW> =
4.	For another frequency		[XEQ] W2	W=?
5.	Repeat step 3 with ω_2.			

Program Listings

Store data

```
01♦LBL "FRE
Q"
02 FIX 2
03 "K1=?"
04 PROMPT
05 STO 01
06 "Z2=?"
07 PROMPT
08 STO 02
09 "N3=?"
10 PROMPT
11 STO 03
12 "Z4=?"
13 PROMPT
14 STO 04
15 "Z5=?"
16 PROMPT
17 STO 05
18 "Z6=?"
19 PROMPT
20 STO 06
21 "W7=?"
22 PROMPT
23 STO 07
24♦LBL "W2"
25 "W=?"
26 PROMPT
27 STO 08
```

$\left[\frac{S^2}{\omega_7^2} + \frac{2Z_6}{\omega_7}\right] + 1$

```
28 RCL 06
29 RCL 07
30 /
31 2
32 *
33 RCL 08
34 *
35 1
36 RCL 08
37 RCL 07
38 /
39 X↑2
40 -
41 R-P
```

$Z_5S + 1$

```
42 RCL 05
43 RCL 08
44 *
45 1
46 R-P
47 XEQ 01
```

$Z_4S + 1$

```
48 RCL 04
49 RCL 08
50 *
51 1
52 R-P
```

S^{N_3}

```
53 XEQ 01
54 RCL 08
55 RCL 03
56 Y↑X
57 *
58 X<>Y
59 RCL 03
60 90
61 *
62 +
63 CHS
64 X<>Y
65 1/X
```

$K_1(Z_2S + 1)$

```
66 RCL 02
67 RCL 08
68 *
69 RCL 01
70 *
71 RCL 01
72 R-P
73 XEQ 01
```

Display

```
74 "G<JW>="
75 ARCL X
76 AVIEW
77 STOP
78 STO 09
79 X<>Y
80 "∡G<JW>=
"
81 ARCL X
82 AVIEW
83 STOP
84 RCL 09
85 LOG
86 "LOG G="
87 ARCL X
88 AVIEW
89 STOP
```

Multiply complex numbers

```
90♦LBL 01
91 X<>Y
92 RDN
93 *
94 RDN
95 +
96 R↑
97 RTN
98 .END.
```

00

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00		50	
	K_1		
	Z_2		
	N_3		
	Z_4		
05	Z_5	55	
	Z_6		
	ω_7		
	ω		
	$\|G(j\omega)\|$		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 010 TOT. REG. 34 USER MODE
ENG ___ FIX 2 SCI ___ ON ___ OFF X
DEG X RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

FREQUENCY RESPONSE OF
A TRANSFER FUNCTION
PROGRAM REGISTERS NEEDED: 25

HEWLETT PACKARD
SOLUTION BOOK:
ELECTRICAL ENGINEERING

ROW 1 (1 - 3)

ROW 2 (3 - 9)

ROW 3 (9 - 15)

ROW 4 (15 - 20)

ROW 5 (21 - 24)

ROW 6 (25 - 34)

ROW 7 (35 - 47)

ROW 8 (47 - 56)

ROW 9 (57 - 68)

ROW 10 (69 - 74)

ROW 11 (74 - 80)

ROW 12 (80 - 86)

ROW 13 (86 - 96)

ROW 14 (97 - 98)

TRANSISTOR AMPLIFIER PERFORMANCE

This program computes certain small-signal properties of a transistor amplifier given the h-parameter matrix and the source and load impedances. Properties computed are: current and voltage gains, and input and output impedances.



Equations:

Definition of h-parameter matrix

$$\begin{bmatrix} v_1 \\ i_2 \end{bmatrix} = \begin{bmatrix} h_i & h_r \\ h_f & h_0 \end{bmatrix} \begin{bmatrix} i_1 \\ v_2 \end{bmatrix}$$

Current gain

$$A_i = \frac{i_2}{i_1} = \frac{-h_f}{1 + h_0 Z_L}$$

Voltage gain

$$A_v = \frac{v_2}{v_1} = \frac{A_i Z_L}{Z_{in}}$$

Voltage gain with source resistor

$$A_{vs} = \frac{v_2}{v_S} = \frac{A_i Z_L}{Z_{in} + Z_S}$$

Input impedance

$$Z_{in} = h_i + h_r Z_L A_i$$

Output impedance

$$Z_{out} = \frac{h_i + Z_S}{h_0 h_i + h_0 Z_S - h_f h_r}$$

Example:

What are the small-signal properties of a transistor which has the following h-parameter matrix and has source and load impedances of 1000 and 10,000 ohms, respectively?

$$[h] = \begin{bmatrix} 1100 & 250E-6 \\ 50 & 25E-6 \end{bmatrix}$$

Keystrokes:	Display:	
[USER]		(set USER mode)
[///][FIX] 2		
[XEQ][ALPHA] SIZE [ALPHA] 023		
0 [ENTER↑] 1100 [ENTER↑] 11		
[XEQ][ALPHA] HP [ALPHA]		
0 [ENTER↑] 250 [EEX] 6 [CHS]		
[ENTER↑] 12 [XEQ][ALPHA] HP [ALPHA]		
0 [ENTER↑] 50 [ENTER↑]		
21 [XEQ][ALPHA] HP [ALPHA]		
0 [ENTER↑] 25 [EEX] 6 [CHS]		
[ENTER↑] 22 [XEQ][ALPHA] HP [ALPHA]		
0 [ENTER↑] 1000 [ENTER]		
0 [ENTER↑] 10000 [XEQ][ALPHA] Z [ALPHA]		
[A]	MAG=-40.00	(A_i)
[R/S]	∠=0.00	
[B]	MAG=-400.00	(A_v)
[R/S]	∠=0.00	
[C]	MAG=-200.00	(A_{vs})
[R/S]	∠=0.00	
[D]	MAG=1,000.00	(Z_{in})
[R/S]	∠=0.00	
[E]	MAG=52,500.00	(Z_{out})
[R/S]	∠=0.00	

User Instructions

SIZE: 023

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program and set USER mode		[USER]	
2	Input h-parameters			
	– Angle	Θ_{ij},deg	[ENTER↑]	
	– Magnitude	h_{ij}	[ENTER↑]	
	– Designation (row, column)	ij	[XEQ] HP	
	Repeat for all 4 parameters			
3	Input termination impedances			
	– Angle of Z_S	Θ_S,deg	[ENTER↑]	
	– Magnitude of Z_S	$\|Z_S\|$	[ENTER↑]	
	– Angle of Z_L	Θ_L,deg	[ENTER↑]	
	– Magnitude of Z_L	$\|Z_L\|$	[XEQ] Z	
4	Compute:			
	– Current gain		[A]	MAG=
			[R/S]	∠=
	– Voltage gain		[B]	MAG=
			[R/S]	∠=
	– Voltage gain with source resistor		[C]	MAG=
			[R/S]	∠=
	– Input impedance		[D]	MAG=
			[R/S]	∠=
	– Output impedance		[E]	MAG=
			[R/S]	∠=

Program Listings

Store matrix

```
01◆LBL "HP"
02 STO 00
03 3
04 -
05 STO 01
06 RDN
07 STO IND
00
08 RDN
09 STO IND
01
10 STOP
```

Store impedances

```
11◆LBL "Z"
12 STO 14
13 RDN
14 STO 15
15 RDN
16 STO 16
17 RDN
18 STO 17
19 CF 05
20 STOP
```

Z_{out}

```
21◆LBL E
22 RCL 19
23 RCL 22
24 RCL 08
25 RCL 11
26 XEQ 01
27 P-R
28 STO 02
29 X<>Y
30 STO 03
31 RCL 19
32 RCL 22
33 RCL 17
34 RCL 16
35 XEQ 01
36 P-R
37 ST+ 02
38 X<>Y
39 ST+ 03
40 RCL 18
41 RCL 21
42 RCL 09
43 RCL 12
44 XEQ 01
45 P-R
46 ST- 02
47 X<>Y
48 ST- 03
49 RCL 08
50 RCL 11
51 RCL 17
52 RCL 16
53 XEQ 02
54 RCL 03
55 RCL 02
56 R-P
57 1/X
58 X<>Y
59 CHS
60 X<>Y
61 XEQ 01
62 GTO 20
```

Voltage gain

```
63◆LBL B
64 XEQ b
65 1/X
66 X<>Y
67 CHS
68 X<>Y
69 RCL 03
70 RCL 02
71 XEQ 01
72 GTO 20
```

Voltage gain with source resistor

```
73◆LBL C
74 XEQ b
75 RCL 17
76 RCL 16
77 XEQ 02
78 1/X
79 X<>Y
80 CHS
81 X<>Y
82 RCL 03
83 RCL 02
84 XEQ 01
85 GTO 20
```

Z_{in}

```
86◆LBL D
87 CF 04
88◆LBL 03
89 SF 05
90 XEQ 00
91 RCL 15
92 RCL 14
93 XEQ 01
94 RCL 09
95 RCL 12
96 XEQ 01
97 RCL 08
98 RCL 11
99 XEQ 02
100 FS? 04
101 RTN
102 GTO 20
```

Program Listings

Current gain

```
103♦LBL A
104 CF 05
105♦LBL 00
106 RCL 19
107 RCL 22
108 RCL 15
109 RCL 14
110 XEQ 01
111 P-R
112 1
113 +
114 R-P
115 RCL 18
116 CHS
117 RCL 21
118 1/X
119 XEQ 01
120 1/X
121 CHS
122 FS? 05
123 RTN
124 GTO 20
```

Multiply complex numbers

```
125♦LBL 01
126 X<>Y
127 RDN
128 *
129 RDN
130 +
131 R↑
132 RTN
```

Add complex numbers

```
133♦LBL 02
134 P-R
135 RDN
136 RDN
137 P-R
138 R↑
139 R↑
140 X<>Y
141 RDN
142 +
143 RDN
144 +
145 R↑
146 R-P
147 RTN
```

Display

```
148♦LBL 20
149 "MAG="
150 XEQ 01
151 X<>Y
152 "∡="
153♦LBL 01
154 ARCL X
155 AVIEW
156 STOP
157 RTN
158♦LBL b
159 SF 05
160 XEQ 00
161 RCL 15
162 RCL 14
163 XEQ 01
164 STO 02
165 X<>Y
166 STO 03
167 SF 04
168 XEQ 03
169 RTN
170 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	Used	50	
	Used		
05		55	
	$\angle h_i$		
	$\angle h_r$		
10		60	
	$\lvert h_i \rvert$		
	$\lvert h_r \rvert$		
	$\lvert R_L \rvert$		
15	$\angle R_L$	65	
	$\lvert R_s \rvert$		
	$\angle R_s$		
	$\angle h_f$		
	$\angle h_0$		
20		70	
	$\lvert h_f \rvert$		
	$\lvert h_0 \rvert$		
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 023 TOT. REG. 62 USER MODE

ENG ___ FIX ___ SCI ___ ON X OFF ___

DEG X RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
04		Return	Continue
05		Return	Continue

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

TRANSISTOR AMPLIFIER
PERFORMANCE
PROGRAM REGISTERS NEEDED: 40

HEWLETT PACKARD
SOLUTION BOOK:
ELECTRICAL ENGINEERING

ROW 1 (1 - 7)

ROW 2 (8 - 15)

ROW 3 (16 - 23)

ROW 4 (23 - 32)

ROW 5 (32 - 39)

ROW 6 (39 - 46)

ROW 7 (47 - 54)

ROW 8 (55 - 63)

ROW 9 (63 - 71)

ROW 10 (72 - 77)

ROW 11 (77 - 85)

ROW 12 (85 - 92)

ROW 13 (93 - 99)

ROW 14 (100 - 106)

ROW 15 (107 - 115)

ROW 16 (116 - 124)

ROW 17 (124 - 135)

ROW 18 (136 - 148)

ROW 19 (148 - 152)

ROW 20 (153 - 160)

ROW 21 (161 - 168)

ROW 22 (169 - 170)

CLASS "A" TRANSISTOR AMPLIFIER BIAS OPTIMIZATION

This program is an automation of the method of bias optimization described in "Designing Class "A" Amplifiers to Meet Specified Tolerances," by Ward J. Helms (Electronics, August 8, 1974). The program requires the user to specify a set of parameters from which it determines, by an iterative technique, the optimum values for R_1, R_2, R_E, and R_L. The minimum power gain is also computed.



Equations:

First, values are specified for the following parameters:

ΔI_{CQ} = maximum desired percentage variation of quiescent current

T_{Amax} = maximum ambient temperature (use the maximum cast temperature for a transistor mounted on a heat sink)

T_{Amin} = minimum ambient temperature

T_{Jmax} = maximum junction temperature rating

P_D = maximum rated power dissipation at 25°C

I_1 = collector current, usually selected for convenience so that I_1 and 10 I_1 bracket the expected operating point

ΔV_{BE} = typical base-emitter voltage change over the range of I_1 to 10 I_1 at 25°C

V_{BE1min} = minimum base-emitter voltage at I_1, 25°C

V_{BE1max} = maximum base-emitter voltage at I_1, 25°C

The transistor's thermal resistance is calculated:

$$\theta_{JA} = (T_{max} - 25°C)/P_D$$

and the minimum load resistance and emitter resistance are estimated:

$$R_{L1} = \frac{\theta_{JA}\ V_{CC}^2}{4.4(T_{Jmax} - T_{Amax})} = R_{Ln}$$

$$R_{E1} = 0.1\ R_{L1} = R_{En}$$

Next, the quiescent, maximum, and minimum collector currents are calculated:

$$I_{CQ} = \frac{V_{CC}}{2(R_{Ln} + R_{En})}$$

$$I_{Cmax} = I_{CQ}(1 + \Delta I_{CQ})$$

$$I_{Cmin} = I_{CQ}(1 - \Delta I_{CQ})$$

From these, we can calculate the base-emitter voltage under hot, high-current conditions (V_{BEX}) and under cold, low-current conditions (V_{BEN}).

$$T_{max} = \theta_{JA}\, I_{CQ}\frac{V_{CC}}{2} + T_{Amax}$$

$$V_{BEX} = V_{BE1min} + \Delta V_{BE}\log\frac{I_{Cmax}}{I_1} - 0.0022(T_{max} - 25°C)$$

$$T_{min} = \theta_{JA}\, I_{CQ}\frac{V_{CC}}{2}\ (1 - (\Delta I_{CQ})^2) + T_{Amin}$$

$$V_{BEN} = V_{BE1max} + \Delta V_{BE}\ \log\frac{I_{Cmin}}{I_1} - 0.0022(T_{min} - 25°C)$$

Now, a better estimate for the emitter resistance can be made:

$$R_{E(n+1)} = \frac{-2(V_{BEX} - V_{BEN})}{I_{Cmax} - I_{Cmin}}$$

From this point, if $V_{BEX} > V_{BEN}$, then R_E is set to zero, R_L is increased by 10%, and the design procedure is repeated. Iterations continue until $\frac{R_{E(n+1)} - R_{En}}{R_{En}} < .5\%$. If at any time the condition $T_{max} > T_{Jmax}$ occurs, R_L is increased by 10%.

When the iterative procedure is complete, T_{max}, I_{Cmax}, T_{min}, and I_{Cmin} are displayed.

Then values for

h_{FEmax} = maximum worst-cast current gain at T_{max} or T_{min} and I_{Cmax} or I_{Cmin}

and

h_{FEmin} = minimum worst-cast current gain at T_{max} or T_{min} and I_{Cmax} or I_{Cmin}

are determined from the transistor's data sheet and the Thevenin-equivalent resistance (R_B) and voltage (V_{BB}) of the amplifier's bias network are calculated:

$$R_B = \frac{h_{FEmax} h_{FEmin} [R_{E(n+1)} (I_{Cmax} - I_{Cmin}) + V_{BEX} - V_{BEN}]}{h_{FEmax} I_{Cmin} - h_{FEmin} I_{Cmax}}$$

$$V_{BB} = V_{BEN} + I_{Cmin} (\frac{R_B}{h_{FEmin}} + R_{E(n+1)})$$

Now the bias resistors R_1 and R_2 are calculated:

$$R_1 = \frac{R_B V_{CC}}{V_{BB}}$$

$$R_2 = \frac{R_B V_{CC}}{(V_{CC} - V_{BB})}$$

Finally, the minimum power gain and minimum signal power are calculated:

$$A_P = \frac{R_B R_L h_{FEmin}}{R_E (R_B + h_{FEmin} R_E)}$$

$$P_S = (1 - \Delta I_{CQ})^2 \left(\frac{{V_{CC}}^2 R_L}{8(R_L + R_E)^2} \right)$$

Example:

A single-stage class "A" amplifier is connected to a 30-V power supply. Calculate the maximum power output and maximum power gain obtained from a transistor over an ambient temperature range of 0°C to 70°C, with a maximum quiescent-current variation of ±20%.

From the transistor's data sheet:

T_{Jmax}	150°C
P_D	= 0.36 W
ΔV_{BE}	= 0.10 v from 3 to 30 mA
V_{BE1min}	= 0.52 v at 3 mA at 25°C
V_{BE1max}	= 0.72 v at 3 mA at 25°C
I_1	= 0.001 A

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 017	
30 [STO] 00	
.2 [STO] 01	
70 [STO] 02	
0 [STO] 03	
150 [STO] 04	
.36 [STO] 05	
.001 [STO] 06	
.1 [STO] 07	
.52 [STO] 08	
.72 [STO] 09	
[XEQ] [ALPHA] BIAS [ALPHA]	T=148.E0
[R/S]	I=18.0E-3
[R/S]	T=74.8E0
[R/S]	I=12.0E-3
[R/S]	H MAX?
600 [R/S]	H MIN?
100 [R/S]	RE=115.E0
[R/S]	RL=888.E0
[R/S]	R2=4.18E3
[R/S]	R1=45.0E3
[R/S]	AP=22.9E0

User Instructions

SIZE: 017

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	Store data			
		V_{CC}	[STO] 00	
		ΔI_{CQ}	[STO] 01	
		T_{Amax}, °C	[STO] 02	
		T_{Amin} °C	[STO] 03	
		T_{Jmax}	[STO] 04	
		P_D	[STO] 05	
		I_1	[STO] 06	
		ΔV_{BE}	[STO] 07	
		V_{BE1min}	[STO] 08	
		V_{BE1max}	[STO] 09	
3	Compute maximum and minimum temperatures			
	and currents		[XEQ] BIAS	T= (max)
			[R/S]	I= (max)
			[R/S]	T= (min)
			[R/S]	I= (min)
			[R/S]	H MAX?
4	Input maximum and minimum h_{FE} to compute	h_{FEmax}	[R/S]	H MIN?
	resistor valves	h_{FEmin}	[R/S]	RE=
			[R/S]	RL=
			[R/S]	R2=
			[R/S]	R1=
	and minimum power gain		[R/S]	AP=

Program Listings

```
 01♦LBL "BIA
S"
```

θ_{JA}

```
 02 ENG 2
 03 RCL 04
 04 25
 05 -
 06 RCL 05
 07 /
 08 STO 05
```

R_{L1}

```
 09 RCL 00
 10 X↑2
 11 *
 12 RCL 04
 13 RCL 02
 14 -
 15 4.4
 16 *
 17 /
 18 STO 13
```

R_{E1}

```
 19 .1
 20 *
 21 STO 14
```

Loop

```
 22♦LBL 00
 23 RCL 00
 24 2
 25 /
 26 ENTER↑
 27 ENTER↑
 28 RCL 13
 29 RCL 14
 30 +
 31 /
 32 STO 10
 33 RCL 05
 34 *
 35 *
 36 RCL 02
 37 +
 38 RCL 04
 39 X<>Y
 40 X>Y?
 41 GTO 05
 42 XEQ 03
 43 CHS
 44 RCL 01
 45 1
 46 +
 47 XEQ 04
 48 RCL 08
 49 +
 50 STO 15
 51 1
 52 RCL 01
 53 X↑2
 54 -
 55 2
 56 /
 57 RCL 10
 58 *
 59 RCL 05
 60 *
 61 RCL 00
 62 *
 63 RCL 03
 64 +
 65 XEQ 03
 66 CHS
 67 1
 68 RCL 01
 69 -
 70 XEQ 04
 71 RCL 09
 72 +
 73 STO 16
 74 RCL 15
 75 X>Y?
 76 GTO 02
 77 -
 78 RCL 10
 79 /
 80 RCL 01
 81 /
 82 RCL 14
 83 X<>Y
 84 STO 14
 85 %CH
 86 .5
 87 X<=Y?
 88 GTO 00
 89 FS? 01
 90 GTO 01
 91 SF 01
 92 GTO 00
```

End Loop

R_B

```
 93♦LBL 01
 94 CF 01
 95 "H MAX?"
 96 PROMPT
 97 "H MIN?"
 98 PROMPT
 99 STO 12
100 X<>Y
101 STO 11
```

Program Listings

```
RCL 01
2
*
RCL 10
*
RCL 14
*
RCL 15
+
RCL 16
-
RCL 12
*
RCL 11
*
1
RCL 01
-
RCL 11
*
1
RCL 01
+
RCL 12
*
-
/
RCL 10
/
STO 02
```

V_{BB}

```
RCL 12
/
RCL 14
+
RCL 19
*
1
RCL 01
-
*
RCL 16
+
STO 03
```

R_1

```
RCL 00
X<>Y
/
RCL 02
*
```

R_2

```
RCL 02
RCL 00
*
LASTX
RCL 03
-
/
```

Display

```
RCL 13
RCL 14
"RE="
XEQ 02
"RL="
XEQ 02
"R2="
XEQ 02
"R1="
XEQ 02
```

Power gain

```
/
RCL 12
*
RCL 02
*
LASTX
RCL 14
RCL 12
*
+
/
LOG
10
*
"AP="
*LBL 02
ARCL X
AVIEW
STOP
RDN
RTN
```

Temperature calculations

```
*LBL 03
FS? 01
XEQ A
25
-
2.2 E-3
*
RTN
*LBL A
"T="
GTO D
```

```
*LBL 04
RCL 10
*
FS? 01
XEQ I
```

Program Listings

```
204 RCL 06
205 /
206 LOG
207 RCL 07
208 *
209 +
210 RTN
211*LBL I
212 "I="
213*LBL D
214 ARCL X
215 AVIEW
216 STOP
217 RTN
218 STO 02
219 0
220 STO 14
221*LBL 05
222 1.1
223 ST* 13
224 GTO 00
225 .END.
```

Current calculations

Display

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	V_{CC}	50	
	ΔI_{CQ}		
	T_{Amin}		
	T_{Amax}		
	T_{Jmax}		
05	P_0 θ_{JA}	55	
	I_1		
	ΔV_{BE}		
	V_{BE1max}		
	V_{BE1min}		
10	I_{CQ}	60	
	h_{FEmax}		
	h_{FEmin}		
	R_{L1}		
	R_{E1}		
15	Used	65	
	Used		
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 017 TOT. REG. 63 USER MODE
ENG 2 FIX ___ SCI ___ ON ___ OFF X
DEG X RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
01		Branch	No-Branch

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

CLASS "A" TRANSISTOR AMPLIFIER
BIAS OPTIMIZATION
PROGRAM REGISTERS NEEDED: 46

HEWLETT PACKARD
SOLUTION BOOK:
ELECTRICAL ENGINEERING

ROW 1 (1 - 4)

ROW 2 (5 - 15)

ROW 3 (16 - 27)

ROW 4 (28 - 40)

ROW 5 (41 - 48)

ROW 6 (49 - 61)

ROW 7 (62 - 70)

ROW 8 (71 - 81)

ROW 9 (82 - 90)

ROW 10 (91 - 95)

ROW 11 (95 - 101)

ROW 12 (102 - 113)

ROW 13 (114 - 126)

ROW 14 (127 - 139)

ROW 15 (140 - 151)

ROW 16 (152 - 160)

ROW 17 (160 - 164)

ROW 18 (164 - 170)

ROW 19 (171 - 181)

ROW 20 (181 - 190)

ROW 21 (190 - 195)

ROW 22 (196 - 202)

ROW 23 (203 - 212)

ROW 24 (212 - 221)

ROW 25 (222 - 225)

ACTIVE FILTER DESIGN

This program computes element values for the standard active filter circuits shown. The user selects corner frequency f_0 or center frequency f_0, midband gain A, peaking factor α, and a capacitor C. The program then prints out a list of elements which form the desired filter.

Equations:

$$\alpha = \frac{1}{Q} = 2\zeta, \text{ where Q is quality factor and } \zeta \text{ is damping factor.}$$

Low pass filter

$$C_5 = C$$

$$C_2 = \frac{4C(A + 1)}{\alpha^2}$$

$$R_1 = \frac{\alpha}{4A\pi f_0 C}$$

$$R_3 = \frac{\alpha}{4\pi f_0 C(A + 1)} = \frac{A}{A + 1} R_1$$

$$R_4 = AR_1$$



High pass filter

$$C_1 = C_3 = C$$

$$C_4 = \frac{C}{A}$$

$$R_2 = \frac{\alpha}{2\pi f_0 C \ (2 + \frac{1}{A})}$$

$$R_5 = \frac{2A + 1}{\alpha 2\pi f_0 C}$$



Bandpass filter

$$C_3 = C_4 = C$$

$$R_1 = \frac{1}{A 2\pi f_0 C\alpha}$$

$$R_2 = \frac{1}{(\frac{2}{\alpha^2} - A)\ 2\pi f_0 C\alpha}$$

$$R_5 = \frac{2}{\alpha 2\pi f_0 C}$$



Example:

Design a high-pass active filter with the following parameters:

f_0 = 10 Hz

A = 10

α = 1

C = 1 μF

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 006	
[XEQ] [ALPHA] AF [ALPHA]	F0?
10 [R/S]	A?
10 [R/S]	PF?
1 [R/S]	C?
1 [EEX] 6 [CHS] [R/S]	1.000 -06
[XEQ] [ALPHA] HP [ALPHA]	C1=C3=1.000E-6
[R/S]	C4=100.0E-9
[R/S]	R2=7.579E3
[R/S]	R5=334.2E3

User Instructions

SIZE: 006

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load Program			
2	Begin execution		[XEQ] AF	FO?
3	Input design specifications:			
	- Corner or center frequency	f_0	[R/S]	A?
	- Midband gain	A	[R/S]	PF?
	- Peaking factor	α	[R/S]	C?
	- Capacitor value	C	[R/S]	
4	Select desired filter characteristic:			
	- Low pass		[XEQ] LP	R1=
			[R/S]	C2=
			[R/S]	R3=
			[R/S]	R4=
			[R/S]	C5=
	- High pass		[XEQ] HP	C1=C3=
			[R/S]	C4=
			[R/S]	R2=
			[R/S]	R5=
	- Band pass		[XEQ] BP	R1=
			[R/S]	R2=
			[R/S]	C3=C4=
			[R/S]	R5=

Program Listings

Initialization

```
01♦LBL "AF"
02 ENG 3
03 "F0?"
04 PROMPT
05 STO 00
06 "A?"
07 PROMPT
08 STO 01
09 "PF?"
10 PROMPT
11 STO 02
12 "C?"
13 PROMPT
14 STO 03
15 STOP
```

Low pass

R_1

```
16♦LBL "LP"
17 RCL 02
18 2
19 /
20 RCL 01
21 /
22 XEQ A
23 /
24 STO 05
25 "R1="
26 XEQ 05
```

C_2

```
27 RCL 03
28 4
29 *
30 RCL 01
31 1
32 +
33 *
34 RCL 02
35 X↑2
36 /
37 "C2="
38 XEQ 05
```

R_3

```
39 RCL 02
40 2
41 /
42 RCL 01
43 1
44 +
45 /
46 RCL 04
47 /
48 "R3="
49 XEQ 05
```

R_4

```
50 RCL 01
51 RCL 05
52 *
53 "R4="
54 XEQ 05
```

C_5

```
55 RCL 03
56 "C5="
57 GTO 05
```

High pass

C_1 & C_3

```
58♦LBL "HP"
59 RCL 03
60 "C1=C3="
61 XEQ 05
```

C_4

```
62 RCL 01
63 /
64 "C4="
65 XEQ 05
```

R_2

```
66 XEQ A
67 RCL 02
68 X<>Y
69 /
70 2
71 RCL 01
72 1/X
73 +
74 /
75 "R2="
76 XEQ 05
```

R_5

```
77 RCL 03
78 RCL 01
79 2
80 *
81 1
82 +
83 RCL 02
84 /
85 RCL 04
86 /
87 "R5="
88 GTO 05
```

Band pass

R_1

```
89♦LBL "BP"
90 XEQ A
91 RCL 01
92 *
93 RCL 02
94 *
95 1/X
96 "R1="
97 XEQ 05
```

R_2

```
98 2
99 RCL 02
100 X↑2
101 /
102 RCL 01
```

Program Listings

```
103 -
104 RCL 04
105 *
106 RCL 02
107 *
108 1/X
109 "R2="
110 XEQ 05
111 RCL 03
112 "C3=C4="
113 XEQ 05
114 2
115 RCL 02
116 /
117 RCL 04
118 /
119 "R5="
120♦LBL 05
121 ARCL X
122 AVIEW
123 STOP
124 RTN
125♦LBL A
126 2
127 PI
128 *
129 RCL 00
130 *
131 RCL 03
132 *
133 STO 04
134 RTN
135 .END.
```

C_3 & C_4

R_5

Display

$2\pi f_0C$

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	f_0	50	
	A		
	α		
	C		
	$2\pi f_0 C$		
05	R1 (LP only)	55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 006 TOT. REG. 40 USER MODE

ENG 3 FIX ___ SCI ___ ON ___ OFF X

DEG X RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

ACTIVE FILTER DESIGN

PROGRAM REGISTERS NEEDED: 35

HEWLETT PACKARD
SOLUTION BOOK:
ELECTRICAL ENGINEERING

ROW 1 (1 - 4)

ROW 2 (5 - 12)

ROW 3 (12 - 18)

ROW 4 (19 - 26)

ROW 5 (26 - 37)

ROW 6 (37 - 45)

ROW 7 (46 - 53)

ROW 8 (53 - 57)

ROW 9 (58 - 60)

ROW 10 (60 - 65)

ROW 11 (66 - 75)

ROW 12 (75 - 84)

ROW 13 (85 - 89)

ROW 14 (89 - 96)

ROW 15 (97 - 107)

ROW 16 (108 - 112)

ROW 17 (112 - 119)

ROW 18 (119 - 128)

ROW 19 (129 - 135)

BUTTERWORTH FILTER DESIGN

This program computes component values for Butterworth filters between equal terminations. Inputs are termination resistance, passband characteristics, and attenuation at some out-of-band frequency.

Before the filter elements can be calculated, a normalized frequency must be computed from the desired cutoff or center frequency and passband characteristics. The normalized frequency is computed by one of these formulas:

Low Pass

$$\omega_n = \frac{\omega}{\omega_0}$$

High Pass

$$\omega_n = \frac{\omega_0}{\omega}$$

Band Pass

$$\omega_n = \frac{\omega^2 - \omega_0^2}{BW\omega}$$

Band Elimination

$$\omega_n = \frac{\omega BW}{\omega_0^2 - \omega^2}$$

The basic form of the filter is this low-pass prototype:



whose elements are given by one of the following sets of formulas:

$$C_i = \frac{1}{\pi f_c R} \sin \frac{(2i - 1)\pi}{2n}, \quad i = 1, 3, 5, \ldots, n - 1$$

$$L_i = \frac{R}{\pi f_c} \sin \frac{(2i - 1)\pi}{2n}, \quad i = 2, 4, 6, \ldots, n$$

where

$$n = \text{INT}\left[\frac{1 + \ln(2 \times 10^{-\Delta dB/10} - 1)}{2\ln(\omega/\omega_0)}\right]$$

Once the low-pass values have been calculated, if some other passband characteristic is desired, the components of the filter are changed by frequency transformation as shown.

PASSBAND CHARACTERISTIC — CIRCUIT ELEMENTS

Low pass

High pass

Band pass

Band elimination



A bit of thought may be necessary to determine whether the L-C's are connected in series or parallel.

NOTE: The program will give erroneous results if asked to compute filter order when ΔdB is small (i.e., when ΔdB ~ Loss (ω_0)).

Example:

Design a 100 Hz wide Butterworth filter centered at 800 Hz with a 30 dB attenuation at 900 Hz. R_0 is 50Ω. The termination resistance R is 50Ω.



Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 016	
[XEQ] [ALPHA] BUT [ALPHA]	R=?
50 [R/S]	FO=?
800 [R/S]	
[XEQ] [ALPHA] BP [ALPHA]	BW=?
100 [R/S]	F1=?
900 [R/S]	A=?
30 [R/S]	N=6.000 E0
[R/S]	1.000 00
	C=16.48 E-6
[R/S]	L=2.402 E-3
[R/S]	2.000 00
	L=112.5 E-3
[R/S]	C=351.7 E-9
[R/S]	3.000 00
	C=61.49 E-6
[R/S]	L=643.6 E-6
[R/S]	4.000 00
	L=153.7 E-3
[R/S]	C=257.5 E-9
[R/S]	5.000 00
	C=45.02 E-6
[R/S]	L=879.2 E-6
[R/S]	6.000 00
	L=41.19 E-3
[R/S]	C=960.8 E-9

User Instructions

SIZE: 016

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	Begin execution		[XEQ] BUT	R=?
		R, Ω	[R/S]	FO=?
		f_0, Hz	[R/S]	
3	Select filter type			
	Low pass		[XEQ] LP	F1=?
		f_1, Hz	[R/S]	A=?
		ΔdB	[R/S]	
	Go to step 4			
	High pass		[XEQ] HP	F1=?
		f_1, Hz	[R/S]	A=?
		ΔdB	[R/S]	
	Go to step 4			
	Band pass		[XEQ] BP	BW=?
		BW	[R/S]	F1=?
		f_1, Hz	[R/S]	A=?
		ΔdB	[R/S]	
	Go to step 4			
	Band elimination		[XEQ] BE	BW=?
		BW	[R/S]	F1=?
		f_1, Hz	[R/S]	A=?
		ΔdB	[R/S]	
4	Answer is displayed automatically			N=
			[R/S]	1.000 00
	(component number is displayed, then			C=
	component value.) Push [R/S] to continue		[R/S]	2.000 00
				L=

Program Listings

Initialization

```
01♦LBL "BUT"
02 ENG 3
03 RAD
04 "R=?"
05 PROMPT
06 STO 05
07 "FO=?"
08 PROMPT
09 XEQ 10
10 STO 07
11 STOP
12♦LBL "LP"
13 1
14 GTO 01
15♦LBL "HP"
16 2
17 GTO 01
18♦LBL "BP"
19 3
20 GTO 01
21♦LBL "BE"
22 4
23♦LBL 01
24 STO 14
25 3
26 X>Y?
27 GTO 00
28 "BW=?"
29 PROMPT
```

Compute filter order

```
30 XEQ 10
31 STO 08
32♦LBL 00
33 "F1=?"
34 PROMPT
35 "A=?"
36 PROMPT
37 10
38 /
39 10↑X
40 2
41 *
42 1
43 -
44 LN
45 STO 12
46 X<>Y
47 XEQ 10
48 XEQ 07
49 RCL 12
50 X<>Y
51 LN
52 ABS
53 /
54 1
55 +
56 2
57 /
58 INT
59 STO 15
60 STO 10
61 "N="
62 ARCL X
63 AVIEW
64 STOP
```

Loop

Butterworth equations

```
65♦LBL 08
66 RCL 15
67 RCL 10
68 -
69 1
70 +
71 STO 09
72 2
73 *
74 1
75 -
76 PI
77 *
78 2
79 /
80 RCL 15
81 /
82 SIN
83 2
84 *
```

display i

$R^{(-1)^i}$

```
85♦LBL 09
86 STO 11
87 RCL 05
88 -1
89 RCL 09
90 PSE
91 Y↑X
92 Y↑X
93 *
```

Frequency transformation

```
94 GTO IND 14
95♦LBL 01
96 RCL 07
97 /
98 XEQ 06
99 GTO 00
100♦LBL 02
```

Program Listings

```
101 RCL 07
102 *
103 1/X
104 XEQ 06
105 CHS
106 GTO 00
107◆LBL 03
108 SF 01
109 RCL 08
110 /
111 XEQ 06
112 XEQ 00
113 ABS
114 1/X
115 RCL 07
116 X↑2
117 /
118 XEQ 06
119 CHS
120 GTO 00
121◆LBL 04
122 SF 01
123 RCL 08
124 *
125 RCL 07
126 X↑2
127 /
128 XEQ 06
129 XEQ 00
130 ABS
131 RCL 07
132 X↑2
133 *
134 1/X
135 XEQ 06
136 CHS
137◆LBL 00
138 "L="
139 X<0?
140 "C="
141 ABS
142 ARCL X
143 AVIEW
144 STOP
145 FS?C 01
146 RTN
147 DSE 10
148 GTO 08
149 RTN
150◆LBL 07
151 STO 11
```

Display (137–146)

```
152 GTO IND
14
153◆LBL 04
154 XEQ 03
155 GTO 00
156◆LBL 02
157 XEQ 01
158◆LBL 00
159 1/X
160 CHS
161 GTO 05
162◆LBL 01
163 RCL 11
164 RCL 07
165 /
166 GTO 05
167◆LBL 03
168 RCL 11
169 X↑2
170 RCL 07
171 X↑2
172 -
173 RCL 11
174 /
175 RCL 08
176 /
177◆LBL 05
178 ABS
179 STO 13
180 RTN
181◆LBL 06
182 -1
183 RCL 09
184 Y↑X
185 *
186 RTN
187◆LBL 10
188 2
189 *
190 PI
191 *
192 RTN
193 .END.
```

Compute normalized frequency (154–158)

Multiply by 2π (187–192)

00	

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00		50	
	temp. storage		
05	R	55	
	ω_0		
	BW		
	i		
10	counter	60	
	ω_n		
	temp. storage		
	$\frac{1}{3}\ \frac{2}{4}$ filter type		
15	n	65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 016 TOT. REG. 58 USER MODE

ENG 3 FIX ____ SCI ____ ON ____ OFF X

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
01		Return	Continue

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

BUTTERWORTH FILTER DESIGN

PROGRAM REGISTERS NEEDED: 43

HEWLETT PACKARD
SOLUTION BOOK:
ELECTRICAL ENGINEERING

ROW 1 (1 - 4)

ROW 2 (4 - 10)

ROW 3 (11 - 15)

ROW 4 (15 - 19)

ROW 5 (20 - 26)

ROW 6 (27 - 32)

ROW 7 (33 - 37)

ROW 8 (38 - 48)

ROW 9 (48 - 59)

ROW 10 (60 - 69)

ROW 11 (70 - 82)

ROW 12 (83 - 94)

ROW 13 (94 - 103)

ROW 14 (104 - 111)

ROW 15 (111 - 119)

ROW 16 (120 - 128)

ROW 17 (129 - 137)

ROW 18 (138 - 145)

ROW 19 (145 - 154)

ROW 20 (154 - 161)

ROW 21 (162 - 173)

ROW 22 (174 - 185)

ROW 23 (186 - 193)

CHEBYSHEV FILTER DESIGN

This program computes component values for Chebyshev filters between equal terminations. Inputs are termination resistance, passband characteristics, attenuation at some out-of-band frequency, and allowable passband ripple.

Before the filter elements can be calculated, a normalized frequency must be computed from the desired cutoff or center frequency and passband characteristics. The normalized frequency is computed by one of these formulas:

Low Pass

$$\omega_n = \frac{\omega}{\omega_0}$$

High Pass

$$\omega_n = \frac{\omega_0}{\omega}$$

Band Pass

$$\omega_n = \frac{\omega^2 - \omega_0{}^2}{BW\omega}$$

Band Elimination

$$\omega_n = \frac{\omega BW}{\omega_0{}^2 - \omega^2}$$

The basic form of the filter is this low-pass prototype:



whose elements are given by one of the following sets of formulas:

$$C_i = \frac{G_i}{2\pi f_c R}, \quad i = 1, 3, 5, \ldots, n$$

$$L_i = \frac{RG_i}{2\pi f_c}, \quad i = 2, 4, 6, \ldots, n - 1$$

where

$$G_1 = \frac{2a_1}{\gamma}$$

$$G_i = \frac{4a_{i-1}(a_i)}{(b_{i-1})(G_{i-1})}, \quad i = 2, 3, 4, \ldots, n$$

$$\gamma = \sinh\left[\frac{\ln\left(\coth \frac{\varepsilon}{40 \log e}\right)}{2n}\right]$$

$$a_i = \sin \frac{(2i - 1)\pi}{2n}, \quad i = 1, 2, 3, \ldots, n$$

$$b_i = \gamma^2 + \sin^2 \frac{i\pi}{n}, \quad i = 1, 2, 3, \ldots, n - 1$$

$$\varepsilon = \left(10^{\Delta dB/10} - 1\right)^{\frac{1}{2}}$$

The filter order is found by using Newton's method to solve for n in the following formula:

$$\left(\omega + \sqrt{\omega^2 - 1}\right)^{2n} + \left(\omega + \sqrt{\omega^2 - 1}\right)^{-2n} = \frac{4}{\varepsilon^2}\left(10^{\Delta dB/10} - 1\right) - 2$$

using

$$n = \frac{\ln\left[\frac{4}{\varepsilon^2}\left(10^{\alpha/10} - 1\right) - 2\right]}{\ln\left(\omega + \sqrt{\omega^2 - 1}\right)}$$

as an initial guess where α is attenuation in dB's.

The resulting value is then increased slightly:

$$n \leftarrow \text{INT}\ (n + 1)$$

Once the low-pass values have been calculated, if some other passband characteristic is desired, the components of the filter are changed by frequency transformation as shown.

PASSBAND CHARACTERISTIC — CIRCUIT ELEMENTS

Low pass



High pass



Band pass



Band elimination



A bit of thought may be necessary to determine whether the L-C's are connected in series of parallel.

NOTE: The program will give erroneous results if asked to compute filter order when ΔdB is small (i.e., when ΔdB ~ Loss (ω_0)).

Example:

Design a low pass Chebyshev filter with the following characteristics:

R = 50Ω f_o = 500 Hz pass band ripple = 3dB, and 30dB attenuation at 600 Hz.

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 013	
50 [STO] 00	
500 [STO] 01	
3 [STO] 03	
30 [STO] 04	
600 [STO] 05	
[XEQ] [ALPHA] CHEB [ALPHA]	TYPE?
1 [R/S]	N=7.000E0
[R/S]	1.000 00
	C=22.40E-6
[R/S]	2.000 00
	L=12.29E-3
[R/S]	3.000 00
	C=29.53E-6
[R/S]	4.000 00
	L=12.79E-3
[R/S]	5.000 00
	C=29.53E-6
[R/S]	6.000 00
	L=12.29E-3
[R/S]	7.000 00
	C=22.40E-6

User Instructions

SIZE: 013

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	Store data:			
	Termination resistance	R	[STO] 00	
	Cutoff frequency	f_0	[STO] 01	
	Band width (if applicable)	BW	[STO] 02	
	Ripple	ΔdB	[STO] 03	
	Attenuation	α	[STO] 04	
	Out-of-Band frequency*	f_1	[STO] 05	
3	Execute program		[XEQ] CHEB	TYPE?
4	Input filter type			
	1-Low pass	1	[R/S]	
	2-High pass	2	[R/S]	
	3-Band pass	3	[R/S]	
	4-Band elimination*	4	[R/S]	
5	Answer is displayed automatically			N=
	(component number is displayed, then		[R/S]	1.000 00
	component value.) Push [R/S] to continue			C=
			[R/S]	2.000 00
				L=
	*f_1 must be within the stop-band for the			
	band elimination filter			

Program Listings

```
 01◆LBL "CHE
B"
```

Convert Hz to r/s

```
 02 ENG 3
 03 RAD
 04 1
 05 XEQ 09
 06 ST* 01
 07 ST* 02
 08 ST* 05
```

Convert dB to ε

```
 09 RCL 03
 10 10
 11 /
 12 10↑X
 13 1
 14 -
 15 SQRT
 16 STO 06
```

Begin to compute filter order

```
 17 RCL 04
 18 10
 19 /
 20 10↑X
 21 1
 22 -
 23 4
 24 *
 25 RCL 06
 26 X↑2
 27 /
 28 2
 29 -
 30 STO 12
```

Filter type

```
 31 "TYPE?"
 32 PROMPT
 33 STO 07
 34 GTO IND
07
```

Compute ω_n

```
 35◆LBL 04
 36 SF 02
 37 XEQ 03
 38 GTO 00
 39◆LBL 02
 40 SF 02
 41 XEQ 01
 42◆LBL 00
 43 1/X
 44 CHS
 45 GTO 05
 46◆LBL 01
 47 RCL 05
 48 RCL 01
 49 /
 50 FS?C 02
 51 RTN
 52 GTO 05
 53◆LBL 03
 54 RCL 05
 55 X↑2
 56 RCL 01
 57 X↑2
 58 -
 59 RCL 05
 60 /
 61 RCL 02
 62 /
 63 FS?C 02
 64 RTN
 65◆LBL 05
 66 ABS
 67 STO 08
 68 ENTER↑
 69 X↑2
 70 1
 71 -
 72 SQRT
 73 +
 74 STO 10
 75 LN
 76 RCL 12
 77 LN
 78 X<>Y
 79 /
 80 STO 09
```

Newton's method root finder

```
 81◆LBL 06
 82 RCL 09
 83 RCL 10
 84 RCL 09
 85 Y↑X
 86 STO 05
 87 ENTER↑
 88 1/X
 89 +
 90 RCL 12
 91 -
 92 RCL 05
 93 ENTER↑
 94 1/X
 95 -
 96 /
 97 RCL 10
 98 LN
 99 /
100 2
```

Program Listings

```
101 /
102 -
103 STO 09
104 LASTX
105 ABS
106 .01
107 X<=Y?
108 GTO 06
109 RCL 09
110 2
111 /
112 1
113 +
114 INT
115 STO 09
116 "N="             Display n
117 ARCL X
118 AVIEW
119 STOP
                     -----------------
120 1
121 STO 11
122 RCL 03           Chebyshev
123 40               setup
124 /
125 1
126 E↑X
127 LOG
128 /
129 ENTER↑
130 +
131 E↑X
132 1
133 X<>Y
134 +
135 LASTX
136 1
137 -
138 /
139 RCL 09
140 STO 10
141 2
142 *
143 1/X
144 Y↑X
145 ENTER↑
146 1/X
147 -
148 2
149 /
150 X↑2
151 STO 08
```

```
152 PI
153 2
154 /
155 RCL 09
156 /
157 SIN
158 STO 12
159 2
160 *
161 RCL 08
162 SQRT
163 /
164♦LBL 10         -----------------
165 STO 05
166 RCL 00          Display i
167 -1
168 RCL 11
169 PSE
170 Y↑X
171 Y↑X
172 *
173 GTO IND         -----------------
07
174♦LBL 01         Frequency
175 RCL 01          transformation
176 /
177 XEQ 11
178 GTO 00
179♦LBL 02
180 RCL 01
181 *
182 1/X
183 XEQ 11
184 CHS
185 GTO 00
186♦LBL 03
187 RCL 02
188 /
189 XEQ 11
190 SF 01
191 XEQ 00
192 1/X
193 RCL 01
194 X↑2
195 /
196 XEQ 11
197 CHS
198 GTO 00
199♦LBL 04
200 RCL 02
201 *
```

Program Listings

202 RCL 01	
203 X↑2	
204 /	
205 XEQ 11	
206 SF 01	
207 XEQ 00	
208 RCL 01	
209 X↑2	
210 *	
211 1/X	
212 XEQ 11	
213 CHS	-----------------
214♦LBL 00	
215 "L="	
216 X<0?	Display L & C
217 "C="	
218 ABS	
219 ARCL X	
220 AVIEW	
221 STOP	
222 FS?C 01	-----------------
223 RTN	
224 DSE 10	
225 GTO 07	
226 RTN	
227♦LBL 07	
228 RCL 12	Loop
229 RCL 09	
230 RCL 10	
231 -	
232 1	Chebyshev
233 +	equations
234 STO 11	
235 1	
236 -	
237 2	
238 *	
239 1	
240 +	
241 PI	
242 *	
243 2	
244 /	
245 RCL 09	
246 /	
247 SIN	
248 STO 12	
249 *	
250 4	
251 *	
252 RCL 05	

253 /	
254 RCL 08	
255 RCL 09	
256 RCL 10	
257 -	
258 PI	
259 *	
260 RCL 09	
261 /	
262 SIN	
263 X↑2	
264 +	
265 /	
266 GTO 10	
267♦LBL 11	
268 -1	
269 RCL 11	
270 Y↑X	
271 *	
272 RTN	-----------------
273♦LBL 09	
274 2	Multiply by
275 *	2π
276 PI	
277 *	
278 RTN	
279 END	

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	R	50	
	ω_o		
	BW		
	Ripple, dB		
	Attenuation		
05	Attenuation freq.	55	
	ε		
	Filter type		
	ω_n		
	n		
10	Counter	60	
	i		
	Temporarty Storage		
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 013 TOT. REG. 62 USER MODE

ENG 3 FIX ___ SCI ___ ON ___ OFF X

DEG ___ RAD X GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
02		Return	Continue
01		Return	Continue

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

CHEBYSHEV FILTER DESIGN

PROGRAM REGISTERS NEEDED: 51

HEWLETT PACKARD
SOLUTION BOOK:
ELECTRICAL ENGINEERING

ROW 1 (1 : 5)

ROW 2 (5 : 12)

ROW 3 (13 : 24)

ROW 4 (25 : 32)

ROW 5 (33 : 40)

ROW 6 (40 : 49)

ROW 7 (50 : 60)

ROW 8 (61 : 72)

ROW 9 (73 : 85)

ROW 10 (86 : 98)

ROW 11 (99 : 108)

ROW 12 (109 : 118)

ROW 13 (119 : 130)

ROW 14 (131 : 143)

ROW 15 (144 : 156)

ROW 16 (157 : 168)

ROW 17 (169 : 178)

ROW 18 (178 : 187)

ROW 19 (188 : 195)

ROW 20 (196 : 205)

ROW 21 (205 : 212)

ROW 22 (212 : 219)

ROW 23 (220 : 229)

ROW 24 (230 : 242)

ROW 25 (243 : 255)

ROW 26 (256 : 267)

ROW 27 (268 : 279)

ROW 28 (279 : 279)

BODE PLOT OF BUTTERWORTH AND CHEBYSHEV FILTERS

This program provides gain, phase and group delay information for Bode plots of n-pole Butterworth or Chebyshev filters. A frequency transformation feature allows four types of filter characteristics: low pass, high pass, band pass, and band elimination. Frequency steps may be either linear (additive Δf) or logarithmic (multiplicative Δf).

The poles of an n-pole Butterworth filter are given by the following expression.

$$s = \sigma_k + j\omega_k = -\sin\left(\frac{2k-1}{n}\frac{\pi}{2}\right) - j\cos\left(\frac{2k-1}{n}\frac{\pi}{2}\right) \qquad (k = 1, \ldots, n)$$

The poles of a Chebyshev filter are derived from Butterworth poles by the following procedure.

Let
$$\beta_k = \frac{1}{n}\sinh^{-1}\frac{1}{\varepsilon}$$

Then the new poles are given by

$$s_k = \sigma_k \sinh \beta_k + j\,\omega_k \cosh \beta_k$$

The gain, phase and delay functions of a filter are given by the following expressions.

The network transfer function is

$$H(j\omega) = \frac{K}{(j\omega - s_1)(j\omega - s_2)\ldots(j\omega - s_n)}$$

$$= \frac{K}{(M_1 \angle \theta_1)(M_2 \angle \theta_2)\ldots(M_n \angle \theta_n)}$$

$$= \frac{K}{M(\omega) \angle \theta(\omega)}$$

in which K is a constant chosen such that

$$|H(j0)| = 1$$

The magnitude of the transfer function is

$$|H(j\omega)| = \frac{K}{\prod_{i=1}^{n} \sqrt{\sigma_i^2 + (\omega - \omega_i)^2}}$$

and its phase is

$$\arg\,[H(j\omega)] = -\theta(\omega) = -\sum_{i=1}^{n} \tan^{-1} \frac{\omega - \omega_i}{-\sigma_1}$$

The normalized group delay is

$$t_g = \frac{d}{d\omega}\{\theta(\omega)\} = \sum_{i=1}^{n} \frac{\sigma_i}{\sigma_i^{\,2} + (\omega - \omega_i)^2}$$

Example:

Plot the response of a 6-pole Butterworth band-pass filter with BW = 100, f_0 = 800. Make a logarithmic plot using steps of $2^{1/8}$ from 400 Hz to 1600 Hz.

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 016	
[SF] 00	
6 [STO] 00	
800 [STO] 02	
100 [STO] 03	
400 [STO] 04	
1600 [STO] 05	
2 √x √x √x [STO] 06	
[XEQ] [ALPHA] BODE [ALPHA]	?
3 [R/S]	0,1-?
1 [R/S]	T=0.027
[R/S]	∡=161.536
[R/S]	MAG=-129.502
[R/S]	F=400.00
[R/S]	T=0.036
[R/S]	∡=158.504
[R/S]	MAG=-121.591
[R/S]	F=436.203
[R/S]	T=0.051
[R/S]	∡=154.506
[R/S]	MAG=-112.727
[R/S]	F=475.683
⋮	⋮

User Instructions

				SIZE: 016
STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program: for Butterworth		[SF] 00	
	for Chebyshev		[CF] 00	
2	Store applicable data:			
	Filter order	n	[STO] 00	
	Pass band ripple	rip	[STO] 01	
	Cutoff frequency	f_o	[STO] 02	
	Band width	BW	[STO] 03	
	Minimum frequency	f_1	[STO] 04	
	Maximum frequency	f_2	[STO] 05	
	Frequency steps	Δf	[STO] 06	
3	Begin execution		[XEQ] BODE	?
4	Input filter type:			
	1-Low pass	1	[R/S]	0,1-?
	2-High pass	2	[R/S]	0,1-?
	3-Band pass	3	[R/S]	0,1-?
	4-Band elimination	4	[R/S]	0,1-?
5	Select: Linear	0	[R/S]	
	Logarithmic	1		
			[R/S]	T=
			[R/S]	∡=
			[R/S]	MAG=
			[R/S]	F=
6	For next increment, continue to press [R/S].			

Program Listings

```
 01◆LBL "BOD
E"
 02 FIX 3
 03 RCL 01          Convert dB
 04 10              ripple to ε
 05 /
 06 10↑X
 07 1
 08 -
 09 SQRT
 10 STO 01
 11 1               Convert Hz to
 12 XEQ 09          r/s
 13 ST* 02
 14 ST* 03
 15 ST* 04
 16 ST* 05
 17 RCL 06
 18 *
 19 STO 15
 20 "?"             Store filter
 21 PROMPT          type
 22 STO 14
 23 "0,1-?"         Set linear or
 24 PROMPT          logarithmic
 25 SF 01           increment
 26 X=0?
 27 CF 01
 28◆LBL E           Initialize
 29 0               registers
 30 STO 09
 31 STO 12
 32 1
 33 STO 11
 34 RCL 00
 35 STO 10
 36 XEQ 07
 37◆LBL 08          Loop
 38 RAD
 39 XEQ 06          Compute s_k
 40 RCL 07
 41 RCL 13
 42 +
 43 RCL 08
 44 R-P
 45 ST/ 11          gain
 46 X<>Y
 47 ST- 12
 48 RCL 08
 49 RCL 07          phase
 50 R-P
 51 ST* 11
```

```
 52 P-R             gain
 53 RCL 13          normalization
 54 +
 55 RCL 08
 56 R-P
 57 X↑2
 58 X<>Y
 59 RDN             time delay
 60 /
 61 ST+ 09
 62 DSE 10
 63 GTO 08
 64 RCL 04
 65 1               frequency
 66 XEQ 09
 67 /
 68 RCL 11          gain, dB
 69 LOG
 70 20
 71 *
 72 RND
 73 RCL 12
 74 1
 75 P-R             phase, degrees
 76 DEG
 77 R-P
 78 CLX
 79 RCL 09
 80 "T="
 81 XEQ 05          Display
 82 "∡="
 83 XEQ 05
 84 "MAG="
 85 XEQ 05
 86 "F="
 87 XEQ 05
 88 RCL 05
 89 RCL 04          Increment
 90 FS? 01
 91 GTO 00
 92 RCL 15
 93 +
 94 GTO 03
 95◆LBL 00
 96 RCL 06
 97 *
 98◆LBL 03
 99 STO 04
100 X<=Y?
101 GTO E
102 RTN
103◆LBL 07
```

Program Listings

```
104 GTO IND          Compute ωn
14
105*LBL 04           BE
106 XEQ 03
107 GTO 00
108*LBL 02           HP
109 XEQ 01
110*LBL 00
111 1/X
112 CHS
113 GTO 04
114*LBL 01           LP
115 RCL 04
116 RCL 02
117 /
118 GTO 04
119*LBL 03           BP
120 RCL 04
121 X↑2
122 RCL 02
123 X↑2
124 -
125 RCL 04
126 /
127 RCL 03
128 /
129*LBL 04
130 STO 13
131 RTN
-----------------------------------
132*LBL 06           Subroutine to
133 FS? 00           compute sk
134 GTO 01
135 XEQ 01
136 1
137 RCL 01
138 1/X
139 R-P
140 X<>Y
141 RDN
142 LASTX
143 +
144 LN
145 RCL 00
146 /
147 E↑X
148 LASTX
149 CHS
150 E↑X
151 +
152 ENTER↑
153 ENTER↑
154 LASTX
155 2
156 *
157 -
158 ST* 08
159 RDN
160 ST* 07
161 2
162 ST/ 08
163 ST/ 07
164 RTN
-----------------------------------
165*LBL 01           Compute
166 RCL 10           Butterworth
167 2                pole location
168 *
169 1
170 -
171 RCL 00
172 /
173 XEQ 09
174 4
175 /
176 1
177 P-R
178 STO 07
179 X<>Y
180 STO 08
181 RTN
-----------------------------------
182*LBL 09           x2π
183 2
184 *
185 PI
186 *
187 RTN
-----------------------------------
188*LBL 05           Display
189 ARCL X
190 AVIEW
191 STOP
192 RDN
193 RTN
194 .END.
```

90	
00	

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	N	50	
	ripple		
	cutoff frequency		
	Band width		
	fmin, ω_1		
05	fmax, ω_2	55	
	Δf		
	$Im\{s_K\}$		
	$Re\{s_K\}$		
	delay		
10	counter	60	
	$\pi\|H(j\omega)\|$		
	$\Sigma\theta(\omega)$		
	ω_n		
	$\frac{1}{3}\,\frac{2}{4}$ filter type		
15	$\Delta\omega$	65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 016 TOT. REG. 56 USER MODE
ENG ____ FIX 3 SCI ____ ON ____ OFF X
DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		Butterworth	Chebyshev
01		Logarithmic	Linear

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

BODE PLOT OF BUTTERWORTH AND CHEBYSHEV FILTERS
PROGRAM REGISTERS NEEDED: 40

HEWLETT PACKARD
SOLUTION BOOK:
ELECTRICAL ENGINEERING

ROW 1 (1 - 4)

ROW 2 (5 - 14)

ROW 3 (14 - 23)

ROW 4 (23 - 28)

ROW 5 (29 - 39)

ROW 6 (39 - 48)

ROW 7 (49 - 60)

ROW 8 (61 - 68)

ROW 9 (69 - 80)

ROW 10 (80 - 84)

ROW 11 (84 - 88)

ROW 12 (89 - 98)

ROW 13 (99 - 106)

ROW 14 (107 - 115)

ROW 15 (116 - 127)

ROW 16 (128 - 136)

ROW 17 (137 - 149)

ROW 18 (150 - 160)

ROW 19 (161 - 171)

ROW 20 (172 - 182)

ROW 21 (183 - 194)

ROW 22 (194 - 194)

TRANSMISSION LINE CALCULATIONS

This program computes the input impedance of a lossy transmission line terminated in Z_L. The program provides an exact solution when the distributed line parameters $R_0 (= \sqrt{L/C})$, R, and G are given, and it provides an approximate solution when R_0, copper loss and dielectric loss are given.



The transmission line shown has a lumped model composed of elements L, C, R, and G. From this model the following equations can be derived:

MODEL



Equations:

$$R_0 = \sqrt{\frac{L}{C}}$$

$$r = \frac{R}{L} = \frac{vR}{R_0}$$

$$g = \frac{G}{C} = v\ R_0 G$$

$$\omega = 2\pi f$$

where

L = inductance/unit length

C = capacitance/unit length

G = conductance/unit length

R = resistance/unit length

v_r = relative phase velocity

$v = 3 \times 10^8 v_r$

f = frequency, Hz

and

$$\alpha = \frac{1}{\sqrt{2}\,v}\left[rg - \omega^2 + \sqrt{(r^2 + \omega^2)(g^2 + \omega^2)}\right]^{1/2}$$

$$\beta = \frac{1}{\sqrt{2}\,v}\left[\omega^2 - rg + \sqrt{(r^2 + \omega^2)(g^2 + \omega^2)}\right]^{1/2}$$

The approximate solution is

$$\mathrm{Re}\{Z_0\} = R_0\left[1 + \frac{1}{2}\left(\frac{\alpha_C - \alpha_D}{\beta_0}\right)\left(\frac{3\alpha_D + \alpha_C}{\beta_0}\right)\right]$$

$$\mathrm{Im}\{Z_0\} = R_0\left[\frac{\alpha_D - \alpha_C}{\beta_0}\right]$$

$$\alpha = \alpha_C + \alpha_D$$

$$\beta = \beta_0\left[1 + \frac{1}{2}\left(\frac{\alpha_C - \alpha_D}{\beta_0}\right)^2\right]$$

where

α_C = Copper loss, nepers/unit length = $\frac{1}{2}\frac{R}{R_0}$

α_D = Dielectric loss, nepers/unit length = $\frac{1}{2}$ GR

$\beta_0 = \frac{\omega}{v}$

Then

$$Z_{in} = Z_0\left(\frac{1 + \Gamma_L e^{-2\gamma \ell}}{1 - \Gamma_L e^{-2\gamma \ell}}\right)$$

where

$$\Gamma_L = \frac{Z_L - Z_0}{Z_L + Z_0}$$

ℓ = line length

Z_L = impedance of termination

Z_0 = characteristic impedance of line = $\mathrm{Re}\{Z_0\} + j\,\mathrm{Im}\{Z_0\}$

γ = propagation constant of line = $\alpha + j\beta$

Z_0 and γ are computed differently depending on which solution is selected.

$$\mathrm{Re}\{Z_0\} = \frac{R_0}{\sqrt{2(g^2 + \omega^2)}} \left[rg + \omega^2 + \sqrt{(r^2 + \omega^2)(g^2 + \omega^2)} \right]^{1/2}$$

$$\mathrm{Im}\{Z_0\} = \frac{\pm R_0}{\sqrt{2(g^2 + \omega^2)}} \left[-(rg + \omega^2) + \sqrt{(r^2 + \omega^2)(g^2 + \omega^2)} \right]^{1/2}$$

in which the + sign is chosen when $g \geqslant r$ and the - sign is chosen when $g < r$.

Example:

A transmission line has the following properties:

$R = 1.2664\ \Omega/\mathrm{cm}$

$G = 0.000\ 041\ 87$ Siemens/cm

$R_0 = 55\ \Omega$

$v_r = 0.85$

What is the input impedance of 3.5 cm of this line at 2 GHz if it is terminated in $Z_L = 75 \angle -30°$?

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 012	
[XEQ] [ALPHA] LINE [ALPHA]	F?
2 [EEX] 9 [R/S]	VR?
0.85 [R/S]	RO?
55 [R/S]	L?
3.5 [R/S]	
[XEQ] [ALPHA] EXACT [ALPHA]	G?
0.00004187 [R/S]	R?
1.2664 [R/S]	ZL?
75 [R/S]	∡ZL?
30 [CHS] [R/S]	∡ZIN=28.48
[R/S]	ZIN=48.01

User Instructions

SIZE: 012

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	Begin execution		[XEQ] LINE	F?
		f, Hz	[R/S]	VR?
		v_r	[R/S]	RO?
		R_0, Ω	[R/S]	L?
		ℓ, cm	[R/S]	
3	For exact solution:		[XEQ] EXACT	G?
		G	[R/S]	R?
		R	[R/S]	ZL?
		Z_L	[R/S]	∡ZL?
		∡Z_L	[R/S]	∡ZIN=
			[R/S]	ZIN=
4	For approximate solution:		[XEQ] APPROX	AD?
		α_D	[R/S]	AC?
		α_C	[R/S]	ZL?
		Z_L	[R/S]	∡ZL?
		∡Z_L	[R/S]	∡ZIN=
			[R/S]	ZIN=

Program Listings

Initialization

```
 01♦LBL "LIN
E"
 02 FIX 2
 03 "F?"
 04 PROMPT
 05 1 E10
 06 /
 07 STO 09
 08 2
 09 PI
 10 *
 11 *
 12 STO 08
 13 "VR?"
 14 PROMPT
 15 STO 02
 16 "RO?"
 17 PROMPT
 18 STO 01
 19 "L?"
 20 PROMPT
 21 STO 11
 22 2
 23 *
 24 3
 25 RCL 02
 26 *
 27 STO 03
 28 /
 29 STO 07
 30 STOP
```

Exact solution

```
 31♦LBL "EXA
CT"
 32 "G?"
 33 PROMPT
 34 "R?"
 35 PROMPT
 36 RCL 03
 37 *
 38 RCL 01
 39 /
 40 STO 05
 41 RDN
 42 RCL 01
 43 *
 44 ST* 03
 45 RCL 08
 46 RCL 05
 47 R-P
 48 SQRT
 49 STO 05
 50 X<>Y
 51 2
 52 /
 53 STO 06
 54 RCL 08
 55 RCL 03
 56 R-P
 57 SQRT
 58 STO 03
 59 X<>Y
 60 2
 61 /
 62 STO 08
 63 RCL 06
 64 +
 65 STO 04
 66 RCL 06
 67 RCL 08
 68 -
 69 STO 02
 70 RCL 05
 71 RCL 03
 72 /
 73 ST* 01
 74 RCL 05
 75 RCL 03
 76 *
 77 ST* 07
 78 GTO "ZIN
"
```

Approximate solution

```
 79♦LBL "APP
ROX"
 80 "AD?"
 81 PROMPT
 82 "AC?"
 83 PROMPT
 84 X<>Y
 85 STO 08
 86 RDN
 87 STO 03
 88 RCL 11
 89 RCL 02
 90 RCL 09
 91 PI
 92 *
 93 1.5
 94 /
 95 X<>Y
 96 /
 97 STO 06
 98 *
```

Program Listings

```
 99 2
100 *
101 STO 07
102 RCL 08
103 10
104 LN
105 20
106 /
107 RCL 06
108 /
109 ST* 03
110 ST* 08
111 RCL 08
112 RCL 03
113 -
114 ENTER↑
115 STO 05
116 RCL 08
117 3
118 *
119 RCL 03
120 +
121 *
122 2
123 /
124 CHS
125 1
126 +
127 R-P
128 ST* 01
129 X<>Y
130 STO 02
131 RCL 05
132 X↑2
133 2
134 /
135 1
136 +
137 RCL 03
138 RCL 08
139 +
140 R-P
141 ST* 07
142 X<>Y
143 STO 04
```

Calculate Zin

```
144♦LBL "ZIN"
145 "ZL?"
146 PROMPT
147 STO 05
148 "∡ZL?"
149 PROMPT
150 STO 06
151 RCL 04
152 RCL 07
153 P-R
154 CHS
155 E↑X
156 STO 07
157 X<>Y
158 180
159 *
160 PI
161 /
162 STO 08
163 RCL 06
164 RCL 02
165 -
166 RCL 05
167 RCL 01
168 /
169 P-R
170 1
171 +
172 R-P
173 1/X
174 -2
175 *
176 X<>Y
177 CHS
178 X<>Y
179 P-R
180 1
181 +
182 R-P
183 RCL 07
184 *
185 1/X
186 X<>Y
187 RCL 08
188 -
189 CHS
190 X<>Y
191 P-R
192 1
193 -
194 R-P
195 1/X
196 2
197 *
198 X<>Y
199 CHS
```

Program Listings

```
200 X<>Y
201 P-R
202 1
203 +
204 R-P
205 RCL 01
206 *
207 X<>Y
208 RCL 02
209 +
210 "∡ZIN="
211 ARCL X
212 AVIEW
213 STOP
214 X<>Y
215 "ZIN="
216 ARCL X
217 AVIEW
218 STOP
219 .END.
```

Display

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00		50	
	R_0		
	v_r		
	$3v_r R_0 G$		
05	$3v_r R_0 G/R$	55	
	$2\ell/3v_r$		
	ω^{I}		
	$f/10^{10}$		
10		60	
	ℓ		
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 012 TOT. REG. 56 USER MODE

ENG ____ FIX 2 SCI ____ ON ____ OFF X

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

TRANSMISSION LINE CALCULATIONS

PROGRAM REGISTERS NEEDED: 45

HEWLETT PACKARD
SOLUTION BOOK:
ELECTRICAL ENGINEERING

ROW 1 (1 - 3)

ROW 2 (4 - 13)

ROW 3 (13 - 19)

ROW 4 (19 - 31)

ROW 5 (31 - 34)

ROW 6 (34 - 44)

ROW 7 (45 - 57)

ROW 8 (58 - 70)

ROW 9 (71 - 78)

ROW 10 (78 - 80)

ROW 11 (80 - 88)

ROW 12 (89 - 99)

ROW 13 (100 - 109)

ROW 14 (110 - 121)

ROW 15 (122 - 133)

ROW 16 (134 - 144)

ROW 17 (144 - 148)

ROW 18 (148 - 158)

ROW 19 (158 - 169)

ROW 20 (170 - 181)

ROW 21 (182 - 194)

ROW 22 (195 - 207)

ROW 23 (208 - 214)

ROW 24 (215 - 219)

TRANSMISSION LINE IMPEDANCE

This program computes high frequency characteristic impedance (Z_0) for five types of transmission lines.

Transmission line configuration	Equation for Z_0
open two-wire line	$Z_0 = \frac{120}{\sqrt{\varepsilon_r}} \ln\left(\frac{2D}{d}\right)$
single wire near ground	$Z_0 = \frac{138}{\sqrt{\varepsilon_r}} \log\left(\frac{4h}{d}\right)$
balanced wires near ground	$Z_0 = \frac{276}{\sqrt{\varepsilon_r}} \log\left\{\frac{2D}{d}\left[1 + \left(\frac{D}{2h}\right)^2\right]^{-\frac{1}{2}}\right\}$
wires in parallel near ground	$Z_0 = \frac{69}{\sqrt{\varepsilon_r}} \log\left\{\frac{4h}{d}\left[1 + \left(\frac{2h}{D}\right)^2\right]^{+\frac{1}{2}}\right\}$
coaxial line	$Z_0 = \frac{60}{\sqrt{\varepsilon_r}} \ln \frac{D}{d}$

Example 1:

Compute Z_0 of RG-218/U coaxial cable. (D = .68 in., d = .195 in., ε_r = 2.3 (polyethylene)).

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 004	
.68 [ENTER↑] .195 [ENTER↑]	
2.3 [XEQ] [ALPHA] [C] [ALPHA]	Z0=49.42

Example 2:

Compute Z_0 of open 2-wire line with D = 6 in., d = .0808 in., ε_r = 1 (air).

Keystrokes:	Display:
6 [ENTER↑] .0808 [ENTER↑] 1 [XEQ] [ALPHA]	
[OP] [ALPHA]	Z0=600.08

Example 3:

Compute Z_0 of an air line consisting of a single .1285 inch wire 6 inches from a ground plane.

Keystrokes:	Display:
.1285 [ENTER↑] 6 [ENTER↑] 1 [XEQ] [ALPHA]	
[SW] [ALPHA]	Z0=313.44

User Instructions

SIZE: 004

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	Compute impedance of open two-wire line.			
	Input: -wire spacing	D	[ENTER↑]	
	-wire diameter	d	[ENTER↑]	
	-relative permittivity	ε_r	[XEQ] OP	Z0=
3	Compute impedance of a single wire near			
	ground. Input:			
	-wire diameter	d	[ENTER↑]	
	-wire height	h	[ENTER↑]	
	-relative permittivity	ε_r	[XEQ] SW	Z0=
4	Compute impedance of balanced wires near			
	ground. Input:			
	-wire spacing	D	[ENTER↑]	
	-wire diameter	d	[ENTER↑]	
	-wire height	h	[ENTER↑]	
	-relative permittivity	ε_r	[XEQ] B	Z0=
5	Compute impedance of wires in parallel			
	near ground. Input:			
	-wire spacing	D	[ENTER↑]	
	-wire diameter	d	[ENTER↑]	
	-wire height	h	[ENTER↑]	
	-relative permittivity	ε_r	[XEQ] P	Z0=
6	Compute impedance of coaxial line. Input:			
	-inside diameter of outer conductor	D	[ENTER↑]	
	-outside diameter of inner conductor	d	[ENTER↑]	
	-relative permittivity	ε_r	[XEQ] C	Z0=

Program Listings

Two wire line

```
♦LBL "OP"
STO 00
RDN
/
2
*
LN
120
*
RCL 00
SQRT
/
GTO 05
```

Single wire near ground

```
♦LBL "SW"
STO 00
RDN
4
*
/
1/X
LOG
138
*
RCL 00
SQRT
/
GTO 05
```

Balanced wires near ground

```
♦LBL B
STO 00
RDN
STO 01
RDN
STO 02
RDN
STO 03
RCL 01
/
2
/
X↑2
1
+
SQRT
1/X
2
*
RCL 03
*
RCL 02
/
LOG
276
*
RCL 00
SQRT
/
GTO 05
```

Wires in parallel near ground

```
♦LBL "P"
STO 00
RDN
STO 01
RDN
STO 02
RDN
STO 03
RCL 01
/
1/X
2
*
X↑2
1
+
SQRT
RCL 01
*
RCL 02
/
4
*
LOG
69
*
RCL 00
SQRT
/
GTO 05
```

Coaxial cable

```
♦LBL C
STO 00
RDN
/
LN
60
*
RCL 00
SQRT
/
```

Display

```
♦LBL 05
FIX 2
"ZO="
ARCL X
AVIEW
```

Program Listings

```
103 STOP
104 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	εr	50	
	h		
	D		
	D (rounded)		
05		55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 004 TOT. REG. 24 USER MODE
ENG ____ FIX 2 SCI ____ ON ____ OFF ____
DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

TRANSMISSION LINE IMPEDANCE

PROGRAM REGISTERS NEEDED: 20

HEWLETT PACKARD
SOLUTION BOOK:
ELECTRICAL ENGINEERING

ROW 1 (1 - 8)

ROW 2 (8 - 14)

ROW 3 (14 - 24)

ROW 4 (25 - 35)

ROW 5 (36 - 48)

ROW 6 (49 - 58)

ROW 7 (58 - 67)

ROW 8 (68 - 80)

ROW 9 (81 - 90)

ROW 10 (91 - 100)

ROW 11 (100 - 104)

NOTES

NOTES

Hewlett-Packard Software

In terms of power and flexibility, the problem-solving potential of the HP-41 programmable calculator is nearly limitless. And in order to see the practical side of this potential, HP has different types of software to help save you time and programming effort. Every one of our software solutions has been carefully selected to effectively increase your problem-solving potential. Chances are, we already have the solutions you're looking for.

Application Pacs

To increase the versatility of your HP-41, HP has an extensive library of "Application Pacs". These programs transform your HP-41 into a specialized calculator in seconds. Included in these pacs are detailed manuals with examples, miniature plug-in Application Modules, and keyboard overlays. Every Application Pac has been designed to extend the capabilities of the HP-41.

You can choose from:

Aviation (Pre-Flight Only) 00041-15018
Clinical Lab 00041-15024
Circuit Analysis 00041-15024
Financial Decisions 00041-15004
Mathematics 00041-15003
Structural Analysis 00041-15021
Surveying 00041-15005
Securities 00041-15026

Statistics 00041-15002
Stress Analysis 00041-15027
Games 00041-15022
Home Management 00041-15023
Machine Design 00041-15020
Navigation 00041-15017
Real Estate 00041-15016
Thermal and Transport Science 00041-15019
Petroleum Fluids 00041-15039

Users' Library

The Users' Library provides the best programs from contributors and makes them available to you. By subscribing to the HP-41 Users' Library you'll have at your fingertips literally hundreds of different programs from many different application areas.

*Users' Library Solutions Books

Hewlett-Packard offers a wide selection of Solutions Books complete with user instructions, examples, and listings. These solution books will complement our other software offerings and provide you with a valuable tool for program solutions.

You can choose from:

Business Stat/Marketing/Sales 00041-90094
Home Construction Estimating 00041-90096
Lending, Saving and Leasing 00041-90086
Real Estate 00041-90136
Small Business 00041-90137
Geometry 00041-90084
High-Level Math 00041-90083
Test Statistics 00041-90082
Antennas 00041-90093
Chemical Engineering 00041-90100
Control Systems 00041-90092
Electrical Engineering 00041-90088
Fluid Dynamics and Hydraulics 00041-90139
Games II 00041-90443

Civil Engineering 00041-90089
Heating, Ventilating & Air Conditioning 00041-90140
Mechanical Engineering 00041-90090
Solar Engineering 00041-90138
Calendars 00041-90145
Cardiac/Pulmonary 00041-90097
Chemistry 00041-90102
Games 00041-90099
Optometry I (General) 00041-90143
Optometry II (Contact Lens) 00041-90144
Physics 00041-90142
Surveying 00041-90141
Time Module Solutions 00041-90395

*Some books require additional memory modules to accomodate all programs.

ELECTRICAL ENGINEERING

RC TIMING
FREQUENCY RESPONSE OF A TRANSFER FUNCTION
TRANSISTOR AMPLIFIER PERFORMANCE
CLASS A TRANSISTOR AMPLIFIER BIAS OPTIMIZATION
ACTIVE FILTER DESIGN
BUTTERWORTH FILTER DESIGN
CHEBYSHEV FILTER DESIGN
BODE PLOT OF BUTTERWORTH AND CHEBYSHEV FILTERS
TRANSMISSION LINE CALCULATIONS
TRANSMISSION LINE IMPEDANCE

HEWLETT PACKARD

00041-90088 Rev. E 1/84 Printed in U.S.A.